[Windows Mobile Themes] Heart1010:
For you out there using iGO and want to have more free ram when navigating here is my little mort script to

1) exit S2U2

2) exit Mobile Shell

or

Sense (script check what you're using)

3) start iGO

4) restart Mobile Shell or Sense regarding what was running before iGO was started

5) restart S2U2

#iGO8 or iGO Primo
#Disable Manila or Mobile Shell then start iGO
#===================================================================

#Replace iGO_Dir with the IGO directory on your PPC
iGO_Dir = "\Storage Card\iGOP\"

#Replace iGO_DirExe with path to IGO executable file
iGO_DirExe = "\Storage Card\iGOP\iGO.exe"

#Replace MobileShell_Dir with the Mobile Shell directory on your PPC
MobileShell_Dir = "\Program Files\Spb Mobile Shell\"

#Replace S2U2_Dir with the S2U2 directory on your PPC
S2U2_Dir = "\Program Files\S2U2\"

#===================================================================

#Show windows cursor (confirmation that the script is started)
ShowWaitCursor

#Check if S2U2 process exist and stop it
If (ProcExists("S2U2.exe")=TRUE)
S2U2_Present = TRUE #Save that S2U2 was running
Run (S2U2_Dir&"iLock2.exe", "-nomsg")
#Kill ("S2U2.exe")
#Kill ("iLock2.exe")
EndIf

#Check if Mobile Shell process exist and stop it
If (ProcExists("MobileShell.exe")=TRUE)
MobileShell_Present = TRUE #Save that MobileShell was running
RunWait(MobileShell_Dir&"MenuLauncher.exe", "-stop")
#Kill ("MobileShell.exe")
EndIf

#Check if Manila Sense / TF3D process exist and stop it
If (ProcExists("manila.exe")=TRUE)
TF3D_Present = TRUE	#Save that Manila/Sense was running
Registro01 = RegKeyExists( "HKLM","\Software\Microsoft\Today\Items\HTC Sense")
Registro02 = RegKeyExists( "HKLM","\Software\Microsoft\Today\Items\TouchFlo 3D")
Registro03 = RegKeyExists( "HKLM","\Software\Microsoft\Today\Items\Sense")

If (Registro01 = TRUE)
RegWriteDWord("HKLM","\Software\Microsoft\Today\Items\HTC Sense","Enabled","00000000")
EndIf
If (Registro02 = TRUE)
RegWriteDWord("HKLM","\Software\Microsoft\Today\Items\TouchFlo 3D","Enabled","00000000")
EndIf
If (Registro03 = TRUE)
RegWriteDWord("HKLM","\Software\Microsoft\Today\Items\Sense","Enabled","00000000")
EndIf

#Redraw Today
Sleep(500)
Show("Desktop")
RedrawToday
Sleep(500)
EndIf

#Start Hibernate.exe to free memory (program must be on iGO directory)
#Run(iGO_Dir&"Hibernate.exe")

#Close Hibernate report windows with OK command
# x = 0
# While (x <= 5)
# Sleep (500)
# OkRequired = WndExists( "Oxios Hibernate 1.40" )
# If (OkRequired = True)
# SendOK()
# x = 6
# EndIf
# x = x + 1
# EndWhile

#Start iGO
RunWait(iGO_DirExe)

#When exit from iGO restart Mobile Shell if it was disabled
If (MobileShell_Present = TRUE)
RunWait(MobileShell_Dir&"MenuLauncher.exe", "-start")
#Run (MobileShell_Dir&"MobileShell.exe")
Sleep (5000)
EndIf

#When exit from iGO restart Manila/Sense if it was disabled
If ((ProcExists("manila.exe")=FALSE) AND (TF3D_Present = TRUE) )
If (Registro01 = TRUE)
RegWriteDWord("HKLM","\Software\Microsoft\Today\Items\HTC Sense","Enabled","00000001")
EndIf
If (Registro02 = TRUE)
RegWriteDWord("HKLM","\Software\Microsoft\Today\Items\TouchFlo 3D","Enabled","00000001")
EndIf
If (Registro03 = TRUE)
RegWriteDWord("HKLM","\Software\Microsoft\Today\Items\Sense","Enabled","00000001")
EndIf
Sleep(5000)
Show("Desktop")
Sleep(5000)
RedrawToday
Sleep (500)
EndIf

#Restart S2U2 if it was disabled
If (S2U2_Present = TRUE)
Run (S2U2_Dir&"iLock2.exe", "-nomsg")
#Sleep (500)
#Run (S2U2_Dir&"S2U2c.exe")
EndIf

Exit

(thx Vostradamus for that menulauncher.exe tip ;)).